Needham, Chapter 6, Figure [36a], p. 317, plot and discussion
November 17, 2016, 1:00 pm PST

In[529]:=
```
(* SCRIPT FOR PICTURE FOR P. 317, NEEDHAM *)

Clear["Global`*"];

(* FUNCTIONS *)

SetAttributes[{v, ptDisk}, Listable];

v[z_] := {Re @ z, Im @ z};

ptDisk[z_] := {FaceForm @ White, EdgeForm @ Black, Disk[v @ z, .01]};

square[z_, scale_] := {Line[ v @ {z + scale .1 I, z + scale (.1 + .1 I)}],
   Line[ v @ {z + scale (.1 + .1 I), z + scale .1}]};


(* Find center and radius,
and euclidean object of hLine through point z with tangent tan at z *)

hLnAtZInMap[z_, ds_] := Module[{s, q, r, x, y},

   If[Re @ ds ≠ 0 ,

    s = Solve[{x, y} ∈ InfiniteLine[v @{z, z + ds}] &&
       {x, y} ∈ InfiniteLine[v @ {-1, 1}], {x, y}];

    q = (x + y I) /. s[[1]]; r = Abs[z - q];
    {q, r, Circle[v @ q, r, {0, Pi}]},

    q = z; r = 1; {q, r, Line[v @ {z - r, z + r}]}]

  ];


(* Given ds, radius of original h-line,
and right intersection with real line, find an equidistant h-Line *)

equiDistLnInMap[z_, rtIS_, ds_, r_] := Module[
   {dsHat, orthoTanRt, tanTupleRt, lftIS, orthoTanLft, tanTupleLft, q, rEq, s, x, y},

   dsHat = ds / Im[z];

   orthoTanRt = I Exp[I (Pi / 2 + dsHat)]; (* i.e. tan to equidistant line *)

   tanTupleRt = v @ { rtIS, rtIS + orthoTanRt};

   lftIS = rtIS - 2 r;

   orthoTanLft = -I Exp[I (Pi / 2 - dsHat)];

   tanTupleLft = v @ {lftIS, lftIS + orthoTanLft};
```

```
    s = Solve[{x, y} ∈ InfiniteLine[tanTupleRt] &&
       {x, y} ∈ InfiniteLine[tanTupleLft], {x, y}];

    q = (x + y I) /. s[[1]];

    rEq = Abs[lftIS - q];

    {q, rEq, Circle[v @ q, rEq, {Arg[rtIS - q], Arg[lftIS - q]}]}

   ];


(* GLOBALS *)

cA = 0; (* center A *)

rA = .5; (* radius A *)

ds = .15 rA; (* arbitrary length *)

z = rA I;

nhLines = 3;

rotAngle = Pi / 5;


(* OBJECTS *)

circU = {Dotted, Circle[v @ 0,  1]};

diamU = Line[v @ {-1, 1}];

diskDS = {LightGray, Disk[v @ z, Abs @ ds]};

lineL = Line[v @ {0, z}];

lineE = {Dashed, Line[v @ {0, (z - ds)}]};

lineds = {Thick, Line[v @ {z,  z - ds}]};

LPrmData = Table[hLnAtZInMap[z, ds Exp[I n rotAngle]], {n,  1,  nhLines, 1}];

LPrmLns = Table[LPrmData[[i, 3]], {i, 1, nhLines, 1}];

dsPts = Table[z - ds Exp[I n rotAngle], {n,  1,  nhLines,  1}];
(* end points of E' lines *)

linesdsRot = Table[{Thick, Line[v @ {z, dsPt}]}, {dsPt,  dsPts}];

EPrmLnData =
  Table[equiDistLnInMap[z, lnL[[1]] + lnL[[2]], ds,  lnL[[2]]], {lnL,  LPrmData}];

EPrmLns = Table[{Dashed, ePrmLn[[3]]}, {ePrmLn,  EPrmLnData}];
```

```
dsLableLn = Line[v @ {z - .15 + .07 I, z - ds / 2}];

sqSymbols = Table[square[Re @ obj[[1]] + obj[[2]], .3], {obj, LPrmData}];


(* COLLECTIONS OF OBJECTS *)

arcs = {diskDS, circU, LPrmLns, EPrmLns};

lines = {diamU, lineL, lineds, lineE, linesdsRot, dsLableLn, sqSymbols};

points = {ptDisk[{0, 1, z}]};


(* LABELS *)

default = 12;
small = 8;

labels = {
   {"0", default, -.05 I},
   {"1", default, 1.05},
   {"z", default, z + .05 I},
   {"L", default, .025 + .15 I},
   {"E", default, -.05 + .15 I},
   {"E'", default, Re @ LPrmData[[3, 1]] + LPrmData[[3, 2]] - .1 + .15 I},
   {"L'", default, Re @ LPrmData[[3, 1]] + LPrmData[[3, 2]] + .025 + .15 I},
   {"𝓡z^π/5", small, Re @ LPrmData[[1, 1]] + LPrmData[[1, 2]] - .05 I},
   {"𝓡z^2 π/5", small, Re @ LPrmData[[2, 1]] + LPrmData[[2, 2]] - .05 I},
   {"𝓡z^3 π/5", small, Re @ LPrmData[[3, 1]] + LPrmData[[3, 2]] - .05 I},
   {"ds", default, z - .15 + .1 I},
   {"Figure 1. Poincaré Disc: Three rotations of
      π/5\nCf. Figure [36a], p. 318, Needham", default + 2, -1.2 I}
  };

text = Table[Text[Style[item[[1]], item[[2]]], v @ item[[3]]], {item, labels}];

Graphics[{arcs, lines, points, text},
 Background → White, PlotRange → All, ImageSize → Large]

labels2 = {
   {"z", default, z },
   {"Fig. 2. Cutout of Fig. 1", default + 2, 1.4 Im @ z I},
   {"ds", default, z - .16 + .08 I}
  };

text2 = Table[Text[Style[item[[1]], item[[2]]], v @ item[[3]]], {item, labels2}];

Graphics[{arcs, lines, points, text2}, Background → White,
 PlotRange → {{-.25, .25}, {.5 - .25, .5 + .25}}, ImageSize → Large]
```

The object is to replot Needham's Figure[36a], p. 316, to correct what may be a mistake in the orientation of $d\hat{s}$ and to illustrate the effects of rotations of the Poincaré disc that take L to a series of L'. The same rotations take the equidistant curve E to a series of E’. The rotations are indicated below the horizontal line in Figure 1. The angles between E and L and their rotated pairs of E’ and L’ were calculated using the equation in paragraph 3, p. 319:

$d\hat{s}$ = ds/Im(z)

It surprised me that he used this equation for the metric on the hyperbolic map, since the object of figure [36a] was to illustrate how one obtains the metric for the Poincaré disc, which we eventually find depends on |z| and not on Im(z). But the figure is apparently plotted on Poincaré Map of the Upper Half Plane (See (43), for example, which mentions the horizon.) In paragraph 4, Needham explains that we can do this because the PD is a Möbius transformation of the map, so it must be conformal.

Construction of the h-lines and equidistant lines

Beginning with the first rotation, $d\hat{s}$ was calculated and L was rotated by Pi/5 and carried to L_1'. The rotation of L was done by rotating ds (at z) by Pi/5 (with ds $e^{iPi/5}$) and finding the center of the new h-line L’ by solving for the intersection of the extension of ds (at z) and the horizon. This is done in the function hLnAtZInMap[z, ds]. The *Mathematica* line that solves for the center is:

```
Solve[{x, y} ∈ InfiniteLine[v @ {z, z+ds}] && {x, y} ∈ InfiniteLine[v @ {-1, 1}], {x, y}];
```

where ∈ is Element rather than Episilon.

Then $d\hat{s}$ was rotated by Pi/2 and a line was constructed (with *Mathematica*) at this angle at the right(ward) point (b) of intersection of L_1' with the horizon. This becomes a tangent line for the equidistant circle E_1'. The tangent line was then reflected about a vertical line and plotted as a second tangent line at the left point (a) of intersection of L with the horizon. Then orthogonals to the tangents at a and b intersect at the center of the equidistant circle, so E_1' can be plotted.

Problem

There appears to be a small error in the calculation of the radii of the equidistant lines or in the angle $d\hat{s}$. The error manifests as a discrepancy between the point $\mathcal{R}_z^{\theta}$(ds) + z and the nearest point of the corresponding equidistant line (See Figure 2). The radius of the equidistant lines appears to be slightly too large, or the angle $d\hat{s}$ is slightly too small. I have not been able to find the error in the source code.